Errata

Document Title: Log Sweep With the 3582A Spectrum Analyzer (AN 245-5)

Part Number: 5952-8805

Revision Date: June 1979

HP References in this Application Note

This application note may contain references to HP or Hewlett-Packard. Please note that Hewlett-Packard's former test and measurement, semiconductor products and chemical analysis businesses are now part of Agilent Technologies. We have made no changes to this application note copy. The HP XXXX referred to in this document is now the Agilent XXXX. For example, model number HP8648A is now model number Agilent 8648A.

About this Application Note

We've added this application note to the Agilent website in an effort to help you support your product. This manual provides the best information we could find. It may be incomplete or contain dated information, and the scan quality may not be ideal. If we find a better copy in the future, we will add it to the Agilent website.

Support for Your Product

Agilent no longer sells or supports this product. You will find any other available product information on the Agilent website:

www.agilent.com

Search for the model number of this product, and the resulting product page will guide you to any available information. Our service centers may be able to perform calibration if no repair parts are needed, but no other support from Agilent is available.

Agilent Technologies

AN 245-5

APPLICATION NOTE 245-5

LOG SWEEP WITH THE HP 3582A SPECTRUM ANALYZER

HEWLETT hp PACKARD



FOREWORD

Spectrum analyzers using the Fast Fourier Transform – like the 3582A – produce a spectrum display with a linear frequency axis. While this is appropriate for narrowband analysis, it is sometimes more informative to view a broadband analysis, such as the entire audio spectrum, with a display that uses a log frequency axis. This form of presentation is generally called "log sweep."

This application note provides a means of modifying the 3582A Spectrum Analyzer operation so that it produces a 256 point log display covering the range of 10 Hz to 25 KHz. The key to this modification is the use of an external controller whose communication path with the 3582A is the HP Interface Bus (HP-IB*). The required control program is given here in both BASIC and HPL languages.

This application note is the fifth of a series on the 3582A. You will find a list of all the notes on the back cover.

*HP-IB is Hewlett-Packard's implementation of IEEE Standard 488 and identical ANSI Standard MC1.1 "Digital interface for programmable instrumentation."

THE HEWLETT-PACKARD MODEL 3582A SPECTRUM ANALYZER



The HP 3582A is a spectrum analyzer covering the frequency range of DC to 25 kHz. Although it is a FFT-based, digital instrument, a special design effort has made it as straightforward to use as a conventional swept analyzer. With dual measurement channels it is possible to measure transfer function gain and phase, as well as the coherence function. A built-in random or pseudo-random noise source, whose spectrum tracks the analysis range, is a useful measurement stimulus. Band Selectable Analysis enables narrowband, high resolution analysis to be applied to any portion of the frequency range. The instrument comes equipped with a flexible HP-IB interface for control and two-way data transfers.

TABLE OF CONTENTS

Section 1:
Introduction

A spectrum presentation with a logarithmically-compressed frequency axis, commonly called "log sweep," is often used in diverse applications such as characterizing audio components, servo response, and acoustics. Part of the advantage of log sweep becomes evident when the amplitude axis is chosen to be logarithmic also. Various parts of the display may then approximate straight line segments whose slopes reveal, to the practiced eye, a good deal about the structure of the device whose spectrum or transfer function is being measured.

By nature, FFT analyzers produce data which is linearly spaced in frequency, and they must be coaxed to convert the data into a log format. With the use of an external controller the 3582A readily adapts to this job. Essentially, the controfller commands the analyzer to generate a large number of data points over the frequency range. From all of these, 256 points can be chosen logarithmically.

The program described in this note generates a good approximation to a true log sweep in this manner: in the range of 10 Hz to 25 KHz, it first calculates 256 frequencies as a geometric series so that they would be uniformly spaced if plotted on log graph paper. Then, in five measurement passes, it collects data from the FFT "bin" nearest to each of the theoretical frequencies, and presents the data points sequentially on the 3582A display CRT, together with a vertical log graticule.

The log display has 256 points whether the 3582A INPUT MODE is single channel or dual channel. (The 3582A normally produces 256 points in single channel mode, and 128 in dual channel.)

Discrete versus continuous spectra.

Because the 256 data points are discrete samples of the entire frequency range, there is a possibility that some isolated spectral lines may be skipped over and hence not be represented in the final log display. For this reason, it is recommended that the program be used with continuous spectra such as those associated with random processes. Measuring transfer functions using the built-in periodic noise source for excitation also works well, since, although the source has a line spectrum, there is one line for every FFT bin in a measurement.

Section 2:
How to Use This Program

This section describes the equipment needed to use the log sweep program (both BASIC and HPL versions). It also discusses measurement setups and control settings of the 3582A.

EQUIPMENT. In addition to a 3582A Spectrum Analyzer and a 98034A HP-IB Interface assembly, the following controller equipment is needed, depending on which program language is to be used:

LANGUAGE	CONTROLLER	ROM FUNCTIONS
BASIC	9835A or 9845A/B desktop computers	plotter, I/O
HPL	9825A calculator	9872A plotter, string, advanced programming, general I/O, extended I/O

Plotter capability is used only for the optional hard-copy plotter output, so you can omit this ROM function if desired.

SETUP AND OPERATION. Most of the control settings of the 3582A are left to the user to set as he chooses, whether manually or through programming. Those which are determined by the program are: 10 dB/div SCALE, RMS AVERAGE (if averaging is requested), SET START MODE, and the SPANs used in each of the five passes.

Before running the program, the test setup should be checked to see whether the sensitivity controls are set satisfactorily so that no overload occurs on any of the measurement spans. This is especially important when using the internal noise source of the 3582A as a stimulus. Its output is constant-power, restricted to the frequency span being measured, rather than constant power-density over all frequencies. Therefore, the narrower spans may exhibit an overload which isn't evident in a 0-25 KHz analysis, for instance.

After loading the program into the controller, you should press RUN. This results in a few seconds of initializing operations, after which some questions are asked. These are listed below, along with comments on their significance. The first question of the pair is the BASIC version, and the second is HPL:

QUESTION	COMMENT
DO YOU WANT RMS AVERAGING? RMS AVERAGING (yes/no)?	For spectra of random processes, RMS averaging provides a means for smoothing the spectral display. This is also helpful for improving signal/noise ratios for transfer function measurements.*
DO YOU WANT NOISE BANDWIDTH COMPENSATION? NOISE BW COMPENSATION (yes/no)?	Because the measurement bandwidth of an FFT analyzer varies with the frequency span, the display of a random source (like white noise) will appear to be stepped unless this compensation is used. It is not needed for transfer functions or for the internal noise source.
SINGLE CHANNEL (1) OR DUAL CHANNEL (2) MEASUREMENT? SINGLE CHAN(1) OR DUAL CHAN(2)?	The program will accept either single or dual input mode. In either case, the information on the display is what is processed; therefore, be sure that the display is what you want. Amplitude displays should use 10 dB/div SCALE. The CRT may have two displays, but the program will process only the higher priority one. See Application Note 245-4 for priority list.

*More discussion of signal averaging, including how the 3582A improves transfer function measurements, may be found in Application Note 245-1, "Signal averaging with the HP 3582A Spectrum Analyzer," part no. 5952-8767.

If you have answered "YES" to the RMS averaging question, the program pauses while prompting you to set the 3582A AVERAGE NUMBER controls to the number of averages you want. After doing this, press CONTINUE and the program will proceed.

During the data collection phase of the program, a message is displayed on line 2 of the 3582A display indicating the pass for which data is currently being gathered. At the end of the five passes, the program processes the accumulated data and presents the final scaled and annotated display on the 3582A. Do not return the analyzer to LOCAL at this point, or the display will be lost.

Before returning for another measurement, the program asks whether you want a hard copy plot made. Answering "YES" activates the plotting routine, which is written primarily for the 7225A Plotter, although the 9872A and other plotters using the "HP-GL" bus language will work also. After you set the diagonal limits as requested, the program draws X and Y scales. It then pauses so that you can manually change the pen color, and then plots the annotation and data.

Pressing CONTINUE once more completes the sequence by clearing the CRT display, returning the 3582A to LOCAL, and beginning the question cycle to start another measurement. If you are making a series of similar measurements, it is not necessary to type in the answers each time, since these controllers retain the previous entries to an "INPUT" or "ent" program line when CONTINUE alone is pressed.

Section 3:
Description of Program Routines

The program is organized as a collection of subroutines which are called up in the proper order by a control sequence labeled "Start." All variables are global rather than local, somewhat simplifying the job of tracing program flow if that becomes necessary. Both BASIC and HPL versions of the program use the same subroutine labels and are organized the same way, as far as possible. Therefore, the following comments apply to either version.

"Setup": This and "Bin_calc" are initializing routines; that is, they only operate when the program is run the first time, and are subsequently bypassed. "Setup" defines the values of a number of tabular arrays and strings. The uses of these are explained in the commentary included with the BASIC listing.

"Bin_calc": Theoretically, this log sweep program operates by selecting signal values at 256 exponentially-spaced frequencies between 10 Hz and 25 KHz. Frequency values are found in the array F. Practically, the program must round the true frequencies to the nearest available FFT "bins" in each of the five spans used in the analysis. This routine does the rounding operation. Both single channel (256 bins) and dual channel (128 bins) operations contribute the same number of points to a given pass. A slight complication occurs in the first pass: for single channel, the START frequency is 10 Hz, while (for arithmetic reasons in the 3582A) it is 9.8 Hz in dual channel. This condition is sensed and dealt with by a (J = 1) term in the formula.

"Data": This is the principle routine of the program. When called, it cycles through the five passes, performing the following in each pass:

a) sets up the frequency range in the 3582A, using tabular data
b) writes the pass identifier message on the display
c) if RMS averaging was indicated, starts averaging and monitors for completion; otherwise waits tabulated time
d) reads binary display data from the 3582A
e) saves only the data from bins selected for that pass
f) subtracts the bandwidth compensation factor, if indicated

Because single and dual channel operations involve different numbers of display points and different memory locations (Application Note 245-4), there is a separate data collection section for each case.

"Sensitivity": In order to display the full-scale measurement sensitivity on the composite log plot, the controller must interrogate the 3582A to determine this item, which is a function of INPUT SENSITIVITY and the AMPLITUDE REFERENCE LEVEL. The routine finds out the kind of display being shown on the 3582A by reading and decoding front panel switch word 77455. It then sends the appropriate HP-IB command and obtains the sensitivity value.

"Display": This writes the composite graphic data, 256 words, on the 3582A display.

"Graticule": Since the internal graticules of the 3582A CRT are linear, the vertical lines cannot conveniently be used for a logarithmic frequency axis. This routine generates a log graticule by drawing a set of 29 vertical lines corresponding to cardinal frequencies. The graticule coexists with the graphic data without interference, since it uses display buffer space reserved for a second 256-point display.

"Annotation": After writing both graphic data and the log graticule on the 3582A display, the final step is to write the alphanumeric labels and messages. These are "pre-recorded" except for part of line 2, which displays the scale sensitivity previously determined by "Sensitivity."

"Hard_copy": This routine reproduces the complete 3582A log display on an HP-IB plotter. It assumes that you are using a blank sheet of paper, since it draws both vertical and horizontal scaling lines. Remember that the plotter must be one that uses "HP-GL" graphics language, such as the 7225A.

Section 4:
Program Modifications

Because of the modular structure of the program, it is a straightforward matter to change it by adding or deleting segments, or modifying the existing coding. This section suggests a couple of possibilities.

Improving transfer function S/N ratio. Over the 3+ decades of frequency covered by this program, some devices you are measuring will exhibit a dynamic range greater than 50 dB. When this is the case, the low amplitude parts of the response may appear noisy. You may be able to improve the S/N ratio in the noisy parts of the response by increasing the gain of Channel B during the corresponding measurement pass. Try the following procedure:

a) Identify the pass(es) for which the gain should be increased.

b) Insert a program line at the beginning of "Data" to increase Channel B gain. For example,

```
IF J=3  THEN OUTPUT      (BASIC)
        Analyzer: "BS8"
if  J=3: wrt "Analyzer".  (HPL)
        "BS8"
```

Be sure to reset the gain to normal for the following passes.

c) Since step (b) changes the transfer function sensitivity, the resulting jump must be subtracted from the collected data, at the rate of 128 per 10 dB. A convenient way to do this is to modify the calculation of "Bwcomp" by a term like

$$-128*(J=3)$$

This procedure was used for the lowpass filter on the cover of the Application Note. Channel B gain was increased 10 dB in pass 5.

Plotting two (or more) traces. The 3582A display can only present two 256-point traces simultaneously, and one of these is used for the log graticule. No such limitation applies to the external hard-copy plotter, however. If you want to display both amplitude and phase on the same graph, for instance, the simplest approach is to make two measurements. After the first has been plotted you should arrange for the program to skip over the graticule and annotation portions of "Hard_copy" and proceed with the graphic output. In the BASIC version, a simple way to do this is to change line 300 to read

```
IF Plot$ = "YES" THEN GOSUB 1620
```

Section 5: Program Listings

BASIC PROGRAM

```
10    ! LOG FREQUENCY "SWEEP" FOR THE 3582A SPECTRUM ANALYZER CONTROLLED
20    ! BY THE 9835A OR 9845A DESKTOP COMPUTER. PROGRAM INCLUDES A
30    ! SUBROUTINE FOR PLOTTING RESULTS ON AN HP-IB X/Y PLOTTER,
40    ! SUCH AS THE 7225A OR 9872A.
50     DIM Pass_ident$(5)[37],Notation$(4)[32]
60     SHORT Incr(5),F(255) ! "F" contains frequencies used in log display.
70     INTEGER Firstbin(5),Lastbin(5),Span(5),Time(5),Bwcomp(5),I,J
80     INTEGER Bin1(255),Bin2(255),Grafic1(255),Grafic2(127),Disp(255),Grat(29)
90    ! "Bin1" and "Bin2" (single and dual channel) contain the numbers of the
100   ! FFT bins collected from each of the 5 spans;  "Grafic1" and "Grafic2"
110   ! are used to save the 3582A display;  "Disp" collects the log-spaced
120   ! display data;  "Grat" contains bin numbers for the log graticule.
130    Analyzer=711
140    GOSUB Setup
150    GOSUB Bin_calc
160 Start: INPUT "DO YOU WANT RMS AVERAGING?",Ave$  ! Main control sequence.
170     INPUT "DO YOU WANT NOISE BANDWIDTH COMPENSATION?",Bwcomp$
180     INPUT "SINGLE CHANNEL (1) OR DUAL CHANNEL (2) MEASUREMENT?",Chan
190       IF Ave$<>"YES" THEN GOTO 220
200     DISP "SET NUMBER OF AVERAGES ON FRONT PANEL; PRESS 'CONTINUE'"
210   PAUSE
220     DISP "COLLECTING DATA"
230     OUTPUT Analyzer;"AV1MD4SC2"
240     GOSUB Data
250     GOSUB Sensitivity
260     GOSUB Display
270     GOSUB Graticule
280     GOSUB Annotation
290     INPUT "DO YOU WANT A HARD COPY PLOT?",Plot$
300     IF Plot$="YES" THEN GOSUB Hard_copy
310     DISP "PRESS 'CONTINUE' FOR ANOTHER MEASUREMENT"
320   PAUSE
```

```
330     RESET Analyzer
340     LOCAL Analyzer
350     GOTO Start                                      ! End of control sequence.
360 Setup:    MAT READ Grat                             ! Load arrays.
370       DATA 0,23,36,45,52,58,63,68,72,75,98,111,120,128,133,138,143
380       DATA 147,150,173,186,195,203,208,214,218,222,225,248,250
390     MAT READ Firstbin                               ! First bin of each of five
400       DATA 0,0,59,113,160,211                       ! segments making up display.
410     MAT READ Lastbin                                ! Last bin of each segment.
420       DATA 0,58,112,159,210,255
430     MAT READ Incr                                   ! Value of FFT bin spacing
440       DATA  0,.2,1,4,20,100                         ! for each segment, 1 chan.
450     MAT READ Span                                   ! HP-IB code for each span.
460       DATA 0,6,8,10,12,14
470     MAT READ Time                                   ! Dwell time before reading
480       DATA 0,7000,4000,2000,1000,1000               ! data, for non-averaging.
490     MAT READ Bwcomp                                 ! Noise bandwidth ratios,
500       DATA 0,0,89,166,256,345                       ! in display units.
510     Pass_ident$(1)="WTA2,    1ST PASS         SPAN   50 Hz "
520     Pass_ident$(2)="WTA2,    2ND PASS         SPAN  250 Hz "
530     Pass_ident$(3)="WTA2,    3RD PASS         SPAN   1 KHz "
540     Pass_ident$(4)="WTA2,    4TH PASS         SPAN   5 KHz "
550     Pass_ident$(5)="WTA2,    5TH PASS         SPAN  25 KHz "
560      Notation$(1)="     10 - 25000 Hz LOG SWEEP      "
570      Notation$(2)=" FULL SCALE SENSITIVITY"
580      Notation$(3)="10       100        1K        10K  "
590      Notation$(4)="          FREQUENCY  Hz            "
600     A=255/(LGT(25000)-LGT(10))                      ! Calculate 256 exponentially
610     FOR I=0 TO 255                                  ! spaced frequencies from 10
620       F(I)=10^(I/A+1)                               ! to 25000 Hz.
630     NEXT I
640   RETURN
650 Bin_calc:                                           ! Determine which bins in
660     FOR J=1 TO 5                                    ! each of the five spans
670       Fr=PROUND(F(Firstbin(J)),0)                   ! are to be saved to make
680       Gr=Fr-.2*(J=1)                                ! up composite log display;
690       K1=Fr MOD Incr(J)                             ! results are stored in
700       K2=Gr MOD (2*Incr(J))                         ! Bin1 (one channel) and
710       FOR I=Firstbin(J) TO Lastbin(J)               ! Bin2 (two channels).
720         X=Incr(J)*PROUND((F(I)-K1)/Incr(J),0)+K1
730         Y=2*Incr(J)*PROUND((F(I)-K2)/(2*Incr(J)),0)+K2
740         Bin1(I)=(X-Fr)/Incr(J)
750         Bin2(I)=(Y-Gr)/(2*Incr(J))
760       NEXT I
770     NEXT J
780   RETURN
790 Display:    OUTPUT Analyzer;"PRS"                   ! Display graphic data.
800       OUTPUT Analyzer USING "16A/256(Y)";"HLTWTM,74400,256",Disp(*)
810   RETURN
820 Graticule:     MAT Grafic1=(-32768)                 ! Assemble and display
830       FOR I=0 TO 28                                 ! log graticule.
840         Grafic1(254-Grat(I))=1023                   ! Grafic1 is borrowed
850       NEXT I                                        ! as a buffer.
860       OUTPUT Analyzer USING "13A/256(Y)";"WTM,74000,256",Grafic1(*)
870   RETURN
880 Data:    FOR J=1 TO 5                               ! Main data collecting and
890       Startfreq=PROUND(F(Firstbin(J)),0)            ! processing routine.
900       OUTPUT Analyzer USING "2A,4D,2A,2D";"AD",Startfreq,"SP",Span(J)
910       OUTPUT Analyzer;Pass_ident$(J)
920       Bwcomp=Bwcomp(J)*(Bwcomp$="YES")
930         IF Ave$<>"YES" THEN GOTO 980
940       OUTPUT Analyzer;"AV2RELST0"                   ! RMS averaging routine.
950       OUTPUT Analyzer;"LST0"                        ! Check for average done.
960         IF BIT(READBIN(Analyzer),6)=0 THEN GOTO 950
970       GOTO 990
980       WAIT Time(J)
990         IF Chan<>2 THEN GOTO 1070
```

```
1000    OUTPUT Analyzer;"HLTLFM,74600,128"          ! Dual channel section.
1010    ENTER Analyzer USING "#,128(Y)";Grafic2(*)
1020      FOR I=Firstbin(J) TO Lastbin(J)
1030        Disp(255-I)=BINAND(Grafic2(127-Bin2(I)),1023)-Bwcomp
1040        IF Disp(255-I)<0 THEN Disp(255-I)=0
1050      NEXT I
1060    GOTO 1130
1070    OUTPUT Analyzer;"HLTLFM,74400,256"          ! Single channel section.
1080    ENTER Analyzer USING "#,256(Y)";Grafic1(*)
1090      FOR I=Firstbin(J) TO Lastbin(J)
1100        Disp(255-I)=BINAND(Grafic1(255-Bin1(I)),1023)-Bwcomp
1110        IF Disp(255-I)<0 THEN Disp(255-I)=0
1120      NEXT I
1130    OUTPUT Analyzer;"RUN"                       ! Ready 3582A for next pass.
1140   NEXT J
1150  RETURN
1160 Sensitivity:  OUTPUT Analyzer;"LFM,77455,1"    ! Determine what kind of
1170    ENTER Analyzer USING "#,Y";X                ! display is being used.
1180    X=BINAND(X,7)
1190      IF X=1 THEN Sens$="LAS"
1200      IF X=2 THEN Sens$="LBS"
1210      IF X=4 THEN Sens$="LXS"
1220    OUTPUT Analyzer;Sens$                       ! Determine sensitivity.
1230    ENTER Analyzer;Sens
1240      IF Sens>200 THEN Sens=9999                ! Default value if
1250  RETURN                                        ! sensitivity uncalibrated.
1260 Annotation:   OUTPUT Analyzer USING "5A,32A";"WTA1,",Notation$(1)
1270    OUTPUT Analyzer USING "5A,23A,4D,5A";"WTA2,",Notation$(2),Sens," dB  "
1280    OUTPUT Analyzer USING "5A,32A";"WTA3,",Notation$(3)
1290    OUTPUT Analyzer USING "5A,32A";"WTA4,",Notation$(4)
1300  RETURN
1310 Hard_copy:                                     ! Routine plots completed log
1320  PLOTTER IS 7,5,"9872A"                        ! display on X-Y plotter.
1330    DISP "SET LOWER LEFT AND UPPER RIGHT LIMITS; PRESS 'ENTER' AFTER EACH"
1340    LIMIT
1350    DISP                                        ! Clear display.
1360    LOCATE 0,100*RATIO,10,90                    ! Set limits of graph area.
1370    SCALE 0,255,0,1023
1380    MOVE 0,0                                    ! Draw 10 dB/division
1390    FOR I=0 TO 8                                ! vertical amplitude scale.
1400      X=(I MOD 2=0)*255
1410      PLOT X,1023*I/8,-1
1420      MOVE X,1023*(I+1)/8
1430    NEXT I
1440    MOVE 0,0                                    ! Draw log-spaced horizontal
1450    FOR I=0 TO 28                               ! frequency scale.
1460      Y=(I MOD 2=0)*1023
1470      PLOT Grat(I),Y,-1
1480      MOVE Grat(I+1),Y
1490    NEXT I
1500    DISP "CHANGE PEN COLOR, IF DESIRED, THEN PRESS 'CONTINUE'"
1510  PAUSE
1520    DISP                                        ! Clear display.
1530    CSIZE 5,.632*RATIO                          ! Plot the four lines
1540    MOVE 0,1090                                 ! of annotation.
1550    LABEL Notation$(1)
1560    MOVE 0,1035
1570    LABEL USING "23A,4D,5A";Notation$(2),Sens," dB  "
1580    MOVE 0,-50
1590    LABEL Notation$(3)
1600    MOVE 0,-100
1610    LABEL Notation$(4)
1620    FOR I=0 TO 255                              ! Plot graph.
1630      PLOT 1,Disp(255-I)
1640    NEXT I
1650    PENUP
1660  RETURN
```

HPL PROGRAM

```
0: "LOG SWEEP FOR THE 3582A AND THE 9825A":
1: dim P$[5,37],A$[3],B$[3],S$[3],I[5],F[0:255],R[5],L[5],S[5],T[5]
2: dim C[5],B[2,0:255],G[0:255],H[0:127],D[0:255],A[0:29],N$[4,32],T$[3]
3: dev "Analyzer",711
4: gsb "Setup"
5: gsb "Bin_calc"
6: "Start":ent "RMS AVERAGING (yes/no)?",A$
7: ent "NOISE BW COMPENSATION (yes/no)?",B$
8: ent "SINGLE CHAN(1) OR DUAL CHAN(2)?",C
9: if A$#"yes";gto 11
10: dsp "SET # AVERAGES; PRESS CONTINUE";stp
11: dsp "COLLECTING DATA"
12: wrt "Analyzer","AV1MD4SC2"
13: gsb "Data"
14: gsb "Sensitivity"
15: gsb "Display"
16: gsb "Graticule"
17: gsb "Annotation"
18: ent "X-Y PLOTTER (yes/no)?",T$;if T$="yes";gsb "Hard_copy"
19: dsp "PRESS CONTINUE FOR NEXT MEAS.";stp
20: clr "Analyzer";lcl "Analyzer";gto "Start"
21: "Setup":0→A[0];23→A[1];36→A[2];45→A[3];52→A[4];58→A[5];63→A[6];250→A[29]
22: 68→A[7];72→A[8];75→A[9];98→A[10];111→A[11];120→A[12];128→A[13];133→A[14]
23: 138→A[15];143→A[16];147→A[17];150→A[18];173→A[19];186→A[20];195→A[21]
24: 203→A[22];208→A[23];214→A[24];218→A[25];222→A[26];225→A[27];248→A[28]
25: 0→R[1];59→R[2];113→R[3];160→R[4];211→R[5]
26: 58→L[1];112→L[2];159→L[3];210→L[4];255→L[5]
27: .2→I[1];1→I[2];4→I[3];20→I[4];100→I[5]
28: 6→S[1];8→S[2];10→S[3];12→S[4];14→S[5]
29: 7000→T[1];4000→T[2];2000→T[3];1000→T[4]→T[5]
30: 0→C[1];89→C[2];166→C[3];256→C[4];345→C[5]
31: "WTA2,    1ST PASS         SPAN   50 Hz "→P$[1]
32: "WTA2,    2ND PASS         SPAN  250 Hz "→P$[2]
33: "WTA2,    3RD PASS         SPAN    1 KHz "→P$[3]
34: "WTA2,    4TH PASS         SPAN    5 KHz "→P$[4]
35: "WTA2,    5TH PASS         SPAN   25 KHz "→P$[5]
36: "     10 - 25000 Hz LOG SWEEP      "→N$[1]
37: ' FULL SCALE SENSITIVITY"→N$[2]
38: "10       100         1K         10K   "→N$[3]
39: "          FREQUENCY HZ               "→N$[4]
40: 255/(log(25000)-log(10))→A
41: for I=0 to 255;int(I/A+1)→F[I];next I
42: ret
43: "Bin_calc":for J=1 to 5
44: prnd(F[R[J]],0)→U;U-.2(J=1)→V
45: I[J]frc(U/I[J])→W;2I[J]frc(V/2I[J])→Z
46: for I=R[J] to L[J]
47: I[J]prnd((F[I]-W)/I[J],0)+W→X
48: 2I[J]prnd((F[I]-Z)/2I[J],0)+Z→Y
49: (X-U)/I[J]→B[1,I];(Y-V)/2I[J]→B[2,I]
50: next I;next J;ret
51: "Display":wrt "Analyzer","PRS"
52: wrt "Analyzer","HLTWTM,74400,256"
53: for I=0 to 255
54: wtb 731,shf(D[I],8);wtb 731,D[I]
55: next I;ret
56: "Graticule":for I=0 to 255;-32768→G[I];next I
57: for I=0 to 28;1023→G[254-A[I]];next I
58: wrt "Analyzer","WTM,74000,256"
59: for I=0 to 255;wtb 731,shf(G[I],8)
60: wtb 731,G[I];next I;ret
```

```
61: "Data":for J=1 to 5
62: prnd(F[R[J]],0)→U
63: fmt 1,c,f4.0,c,f2.0
64: wrt "Analyzer.1","AD",U,"SP",S[J]
65: wrt "Analyzer",P$[J];C[J](B$="yes")→B
66: if A$#"yes";gto 71
67: wrt "Analyzer","AV2RELST0"
68: wrt "Analyzer","LST0"
69: if bit(6,rdb("Analyzer"))=0;gto -1
70: gto 72
71: wait T[J]
72: if C#2;gto 80
73: wrt "Analyzer","HLTLFM,74600,128";red "Analyzer"
74: for I=0 to 127;rdb(731)→X;rdb(731)→Y
75: ior(shf(X,-3),Y)→H[I];next I
76: for I=R[J] to L[J]
77: band(H[127-B[2,I]],1023)-B→D[255-I]
78: if D[255-I]<0;0→D[255-I]
79: next I;gto 87
80: wrt "Analyzer","HLTLFM,74400,256";red "Analyzer"
81: for I=0 to 255;rdb(731)→X;rdb(731)→Y
82: ior(shf(X,-8),Y)→G[I];next I
83: for I=R[J] to L[J]
84: band(G[255-B[2,I]],1023)-B→D[255-I]
85: if D[255-I]<0;0→D[255-I]
86: next I
87: wrt "Analyzer","RUN"
88: next J;ret
89: "Sensitivity":wrt "Analyzer","LFM,77455,1"
90: rdb("Analyzer")→X;band(rdb("Analyzer"),7)→X
91: if X=1;"LAS"→S$
92: if X=2;"LBS"→S$
93: if X=4;"LXS"→S$
94: wrt "Analyzer",S$;red "Analyzer",S
95: if S>200;9999→S
96: ret
97: "Annotation":fmt 1,c,c;fmt 2,c,c,f4.0,c
98: wrt "Analyzer.1","WTA1,",N$[1]
99: wrt "Analyzer.2","WTA2,",N$[2],S," dB  "
100: wrt "Analyzer.1","WTA3,",N$[3]
101: wrt "Analyzer.1","WTA4,",N$[4];ret
102: "Hard_copy":psc 705;fxd 0
103: dsp "SET LIMITS THEN PRESS CONTINUE";stp
104: scl 0,255,-100,1123;plt 0,0,1
105: for I=0 to 8;(frc(I/2)=0)255→X
106: plt X,1023I/8,2;plt X,1023(I+1)/8,1
107: next I;plt 0,0,1
108: for I=0 to 28;(frc(I/2)=0)1023→Y
109: plt A[I],Y,2;plt A[I+1],Y,1;next I
110: dsp "CHANGE PEN; PRESS CONTINUE";stp
111: csiz 2.5,1.18;plt 0,1090,1;lbl N$[1]
112: plt 0,1035,1;lbl N$[2],S," dB  "
113: plt 0,-50,1;lbl N$[3]
114: plt 0,-100,1;lbl N$[4]
115: for I=0 to 255;plt I,D[255-I];next I
116: pen;ret
*15830
```

3582A SPECTRUM ANALYZER
APPLICATION NOTES

NUMBER	TITLE	DESCRIPTION
245-1	**SIGNAL AVERAGING WITH THE HP 3582A SPECTRUM ANALYZER**	Two kinds of digital averaging are available in the 3582A. This note explains the distinctions, features, and applications of each.
245-2	**MEASURING THE COHERENCE FUNCTION WITH THE HP 3582A SPECTRUM ANALYZER**	A somewhat unfamiliar, though very useful signal measure is available from the 3582A: the coherence function. Its properties and two important applications are explained.
245-3	**THIRD OCTAVE ANALYSIS WITH THE HP 3582A SPECTRUM ANALYZER**	The 3582A may be used to make this traditional measurement through the use of an external HP-IB controller. The note provides some background material and discusses compliance with the ANSI standard. Complete programs in both BASIC and HPL are included.
245-4	**ACCESSING THE 3582A MEMORY WITH HP-IB**	Faster data transfers and the cross-power spectrum are examples of the advantage of having access to the 3582A Read/Write memory. The note includes memory maps, data formats, and several illustrative BASIC programs.
245-5	**LOG SWEEP WITH THE HP 3582A SPECTRUM ANALYZER**	This note explains a way to use an external controller to produce a log frequency axis (log sweep) on the 3582A display, spanning more than 3 decades. The annotated program is listed in both BASIC and HPL versions.

HEWLETT hp PACKARD

June 1979

For more information, call your local HP sales office or East (301) 948-6370 • Midwest (312) 255-9800 • South (404) 955-1500 • West (213) 877-1282; Or write, Hewlett-Packard, 1501 Page Mill Rd., Palo Alto, California 94304.
In Europe, Hewlett-Packard S.A., 7, rue du Bois-du-lan, P.O. Box, CH-1217 Meyrin 2-Geneva, Switzerland. **In Japan,** Yokogawa-Hewlett-Packard Ltd., 29-21 Takaido-Higashi 3-chome, Suginami-ku, Tokyo 168.

Printed in U.S.A.

5952-8805